Esempi riassuntivi seguenti

- **In ognuno degli esempi seguenti, sia Esempio*N* (*N* va da 1 a 14):**
  - **versione C++: programma esempio*N*, due file es*N*a.cpp ed es*N*b.cpp;**
  - **versione equivalente Assembler: due file es*N*a.s ed es*N*b.s**
- **Si può ottenere un programma eseguibile equivalente (nel file *a.out*) in uno di questi quattro modi possibili:**

```
g++  esNa.cpp esNb.cpp
g++  esNa.cpp esNb.s
g++  esNa.s esNb.cpp
g++  esNa.s esNb.s
```

Esempio 1: somma con variabili globali *alfa* e *beta* (1)

```
// programma sommmaintGlob, file es1a.cpp
#include"servi.cpp"
extern "C" int elab1(int n, int m);
int alfa, beta;
int main()
{     int ris;
      alfa = leggiint(); beta = leggiint();
      ris = elab1(alfa, beta);
      scriviint(ris); nuovalinea();
      return 0;
};

// programma sommmaintGlob, file es1b.cpp
extern "C" int elab1(int n1, int n2)
{     int i, j;
      i = n1+n2;
      j = n1-n2;
      return i*j;
};
```

Esempio 1: le variabili globali vengono indirizzate da *main()* tramite RIP (2)

```
# programma sommaintGlob, file es1a.s
# registri per i parametri: rdi, rsi
# registro per il  risultato: rax
.include "servi.s"
.data
alfa:       .long       0
beta:       .long       0

.text
# .extern elab1
.global main
.set  ris, -8
main:
      pushq %rbp                # prologo
      movq  %rsp, %rbp
      subq   $8, %rsp
      # ris: 4 byte (allineamento a 8)

      call     leggiint
      movl   %eax, alfa(%rip)  # risultato  in alfa
      call     leggiint
      movl   %eax, beta(%rip) # risultato in beta

      movl   alfa(%rip), %edi
      movl   beta(%rip), %esi
      call     elab1
      movl   %eax, ris(%rbp)  # risultato in ris

      movl   ris(%rbp), %edi   # parametro ris
      call     scriviint
      call     nuovalinea

      movl  $0, %eax

      leave                          # epilogo
      ret
```

Esempio 1: somma con variabili globali, ricopiate da *elab1()* in *n1*, *n2* e indirizzate tramite RBP (3)

```
# programma somimaintGlob, file es1b.s
# registri per i parametri: rdi, rsi
# registro per il  risultato: rax
.text
.global elab1
.set  i, -16
.set  j, -12
.set  n1, -8
.set  n2, -4
elab1:
      pushq %rbp                    # prologo
      movq  %rsp, %rbp
      subq    $16, %rsp
      # i: 4 byte, j: 4 byte, n1: 4 byte, n2: 4byte

      movl   %edi, n1(%rbp)
      movl   %esi, n2(%rbp)

      movl   n1(%rbp), %eax
      addl    n2(%rbp), %eax
      movl   %eax, i(%rbp)          # i = n1+ n2

      movl   n1(%rbp), %eax
      subl    n2(%rbp), %eax
      movl   %eax, j(%rbp)          # j = n1- n2

      movl   i(%rbp), %eax          # ritorna i * j
      imull   j(%rbp), %eax

      leave                         # epilogo
      ret
```

Esempio 2: somma con variabili locali *a*, *b* e *ris* (1)

```
// programma sommaintLoc, file es2a.cpp
#include"servi.cpp"
extern "C" int elab2(int n, int m);
int main()
{     int a, b, ris;
      a = leggiint(); b = leggiint();
      ris = elab2(a, b);
      scriviint(ris); nuovalinea();
      return 0;
};

// programma sommaintLoc, file es2b.cpp
extern "C" int elab2(int n1, int n2)
{     int i, j;
      i = n1+ n2;
      j = n1- n2;
      return i*j;
};
```

Esempio 2: somma con variabili locali ricopiate da *main()* nei registri, utilizzando RBP (2)

```
# programma sommaintLoc, file es2a.s
# registri per i parametri: rdi, rsi
# registro per il  risultato: rax
.include "servi.s"
.text
# .extern elab2
.global main
.set  a, -12
.set  b, -8
.set  ris, -4
main:
      pushq %rbp                # prologo
      movq  %rsp, %rbp
      subq   $16, %rsp
      # a: 4 byte, b: 4 byte,ris: 4 byte,
      # allineamento

      call     leggiint
      movl   %eax, a(%rbp)   # risultato in a
      call     leggiint
      movl   %eax, b(%rbp)   # risultato in b

      movl    a(%rbp), %edi
      movl    b(%rbp), %esi
      call     elab2
      movl    %eax, ris(%rbp) # risultato  in ris

      movl    ris(%rbp), %edi  # parametro ris
      call     scriviint
      call     nuovalinea

      movl   $0, %eax

      leave                     # epilogo
      ret
```

Esempio 2: somma con variabili locali

```
# programma sommaintLoc, file es2b.s
# registri per i parametri: rdi, rsi
# registri per il  risultato: rax
.text
.global elab2
.set  i, -16
.set  j, -12
.set  n1, -8
.set  n2, -4
elab2:
      …
      # come il file  elab1.s
      …
     leave
     ret
```

Esempio 3: somma con parametro riferimento (1)

```
// programma sommaintRif, file es3a.cpp
#include"servi.cpp"
extern "C" void elab3(int& tot, int n1, int n2);
int main()
{     int a, b; int& ris;
      a = leggiint(); b = leggiint();
      elab3(ris, a, b);
      scriviint(ris); nuovalinea();
};

// programma sommaintRif, file es3b.cpp
extern "C" void elab3(int& tot, int n1, int n2)
{     int i, j;
      i = n1+n2;
      j = n1-n2;
      tot = i*j;
};
```

Esempio 3: somma con parametro formale riferimento *tot* e parametro attuale riferimento *ris* (locale a *main()*) (2)

```
# programma sommaintRif, file es3a.s
# registri per i parametri: rdi, rsi, rdx
# registro per il  risultato: rax
.include  "servi.s"
.text
# .extern   elab3
.global   main
.set  a, -16
.set  b, -12
.set  ris, -4     # è un int
                  # trasmesso a elab3 con LEA
main:
      pushq %rbp              # prologo
      movq  %rsp, %rbp
      subq   $16, %rsp
      # a: 4 byte, b: 4 byte, ris:  8 byte

      call     leggiint
      movl   %eax, a(%rbp)
      call     leggiint
      movl   %eax, b(%rbp)
```

```
      leaq    ris(%rbp), %rdi  # parametro &ris
      movl   a(%rbp), %esi    # parametro a
      movl   b(%rbp), %edx   # parametro b
      call     elab3
      # ris: intero per il risultato
      # riferimento attuale: indirizzo di ris
      movl    ris(%rbp), %edi
      call      scriviint
      call      nuovalinea

      movl    $0, %eax

      leave                          # epilogo
      # muore ris,
      ret
```

Esempio 3: somma con parametro riferimento (3)

```
# programma sommaintRif, file es3b.s
# registri per i parametri: rdi, rsi, rdx
# registro per il  risultato: rax
.text
.global elab3
.set  i, -24
.set  j, -20
.set  tot, -16                # int&
.set  n1, -8
.set  n2, -4
elab3:
      pushq %rbp              # prologo
      movq  %rsp, %rbp
      subq   $24, %rsp
      # i: 4 byte, j: 4 byte
      # tot: 8 byte; i, j, n1, n2 : 4 byte
      # in %rdi  c'è  l'indirizzo di ris,
      # che va in tot
      movq  %rdi, tot(%rbp)
      movl   %esi,  n1(%rbp)
      movl   %edx, n2(%rbp)
```

```
      movl    n1(%rbp), %eax
      addl    n2(%rbp), %eax
      movl     %eax, i(%rbp)      # i = n1+n2
      movl     n1(%rbp), %eax
      subl      n2(%rbp), %eax
      movl      %eax, j(%rbp)     # j = n1-n2

      movq    tot(%rbp), %rdx   # tot: contiene ris&
      movl     i(%rbp), %eax
      imull    j(%rbp), %eax
      movl     %eax, (%rdx)
      # risultato i*j va nella variabile riferita da tot
      # (tot) = i*j

      leave                           # epilogo
     # muoiono n1, n2, i, j, tot, sopravvive (tot)
      ret
```

Esempio 4 : parametro puntatore (1)

```
// programma puntatore, fle es4a.cpp
#include "servi.cpp"
extern "C" void add(int* p, int i);
int main()
{     int a, b;
      a = leggiint(); b = leggiint();
      add(&a, b);
      scriviint(a); nuovalinea();
      return 0;
};

// programma puntatore, file es4b.cpp
extern "C" void add (int* p, int i)
{     *p = *p + i;          // attenzione: non p = p+i, ma *p = *p + i
                            // aritmetica degli interi e non aritmetica degli indirizzi
};
```

Esempio 4 : parametro puntatore (2)

```
# programma puntatore, file es4a.s
# registri per i parametri: rdi, rsi
.include "servi.s"
.text
# .extern add
.global main
.set  a, -8
.set  b, -4
main:
      pushq %rbp            # prologo
      movq  %rsp, %rbp
      subq   $8, %rsp
      # a: 4 byte, b: 4 byte

      call     leggiint
      movl   %eax, a(%rbp)
      call     leggiint
      movl   %eax, b(%rbp)
```

```
      leaq     a(%rbp), %rdi   # &a
      movl    b(%rbp), %esi
      call      add
      movl     a(%rbp), %edi
      call      scriviint
      call      nuovalinea

      movl     $0, %eax

      leave                    # epilogo
      ret
```

Esempio 4 : parametro puntatore (3)

```
# programma puntatore, file es4b.s
# registri per i parametri: rdi, rsi
.text
.global add
.set  p, -16
.set  i, -8
add:
      pushq %rbp              # prologo
      movq  %rsp, %rbp
      subq   $16, %rsp
      # p: 8 byte, i: 4 byte

      movq  %rdi, p(%rbp)
      movl   %esi, i(%rbp)

      movq  p(%rbp), %rax
      movl   (%rax), %ebx
      addl    i(%rbp), %ebx
      movl   %ebx, (%rax)

      leave                   # epilogo
      ret
```

```
# programma puntatore, file es6b.s
# versione semplificata
# registri per i parametri: rdi, rsi
.text
.global add
add:
            movq   %rdi, %rax
            movl          (%rax), %ebx
            addl          %esi, %ebx
            movl          %ebx, (%rax)

            ret
```

Esempio 5: parametro riferimento di puntatore (1)

```
// programma puntatoreRif, file es5a.cpp
#include "servi.cpp"
extern "C" void trovamin(int*& p, int* pa, int* pb);
// restituisce in p uno fra pa e pb, a seconda del minimo intero puntato
int main()
{     int n, m; int* pun;
      n = leggiint();
      m = leggiint();
      trovamin(pun, &n, &m);
      scriviint(*pun); nuovalinea();
      return 0;
};

// programma puntatoreRif, file es5b.cpp
extern "C" void trovamin(int*& p, int* pa, int* pb)
{ if (*pa <= *pb) p = pa; else p = pb;
}
```

Esempio 5: parametro riferimento di puntatore (2)

```
// programma puntatoreRif, file es5a.s
# registri per i parametri: rdi, rsi, rdx
# registri per il  risultato: rax
.include "servi.s"
.text
# .extern trovamin
.global main
.set  n, -16
.set  m, -12
.set  pun, -8
main:
      pushq %rbp                  # prologo
      movq  %rsp,  %rbp
      subq   $16, %rsp
      # n: 4 byte, m: 4 byte, pun: 8 byte
      call     leggiint
      movl   %eax, n(%rbp)
      call     leggiint
      movl   %eax, m(%rbp)
```

```
      leaq   pun(%rbp), %rdi  # &pun
      leaq   n(%rbp), %rsi      # &n
      leaq   m(%rbp), %rdx    # &m
      call     trovamin
      # pun (*p) contiene & n o &m
      movq  pun(%rbp), %rax
      movl   (%rax), %edi       # pun vale &m o &n
      call     scriviint            # *pun vale m o n
      call     nuovalinea

      movl   $0, %eax

      leave                          # epilogo
      ret
```

Esempio 5: parametro riferimento di puntatore (3)

```
# programma puntRif, file es5b.s
# registri per i parametri: rdi, rsi, rdx
.text
.global trovamin
.set  p, -24     # p: riferimento di puntatore
.set  pa, -16
.set  pb, -8
trovamin:
      pushq %rbp            # prologo
      movq  %rsp, %rbp
      subq   $24, %rsp
      # p: 8 bye, pa: 8 byte, pb: 8 byte

      movq  %rdi, p(%rbp)
      movq  %rsi, pa(%rbp)
      movq  %rdx, pb(%rbp)

      movq  pa(%rbp), %rdi
      movl   (%rdi), %esi
      movq  pb(%rbp), %rdi
      movl   (%rdi), %edx
      cmpl   %esi, %edx
      jp       oltre          # edx (*pb) > esi (*pa)
```

```
            movq   p(%rbp), %rdi
            movq  pa(%rbp), %rsi
            movq   %rsi, (%rdi)      # pa va in *p
            jmp     avanti
oltre:      movq  p(%rbp), %rdi
            movq  pb(%rbp), %rsi
            movq  %rsi, (%rdi)       # pb va in *p

avanti:  leave                        # epilogo
            ret

           # *p (ossia pun) contiene pa o pb
```

Esempio 6: parametro array (1)

```
// programma array, file es6a.cpp
#include"servi.cpp"
extern "C" void raddoppia(int a[], int n);
int main()
{     int ar[5]; int i;
      for(i=0; i<5; i++) ar[i] = leggiint();
      raddoppia(ar, 5);
      for(i=0; i<5; i++) scriviint(ar[i]);
      nuovalinea();
};

// programma array, file es6b.cpp
extern "C" void raddoppia(int a[], int n)
{     int i;
      for(i=0; i<n; i++) a[i] = 2*a[i];
};
```

Esempio 6: parametro array (2)

```
# programma array, file es6a.s
# registri per i parametri: rdi, rsi
.include "servi.s"
.text
# .extern raddoppia
.global main
.set  ar, -24
.set  i, -4
main:
        pushq       %rbp                    # prologo
        movq        %rsp, %rbp
        subq $24, %rsp
        # a[5]:  5 int = 20 byte, i: 4 byte

        movl $0,  i(%rbp)                   # i = 0
rip1: cmpl $5, i(%rbp)
        jge    avan1                        # salta se i>=5

        call   leggiint
        movslq i(%rbp), %rdi
        movl %eax, ar(%rbp,%rdi,4) # lettura di a[i]
        incl   i(%rbp)                      # i++
        jmp    rip1
```

```
avan1:  leaq    ar(%rbp), %rdi  # &ar[0]
        movl    $5, %esi
        call    raddoppia

        movl    $0, i(%rbp)         # i = 0
rip2:   cmpl    $5, i(%rbp)         # scritture
        jge     avan2
        movslq  i(%rbp), %rsi
        movl    ar(%rbp, %rsi,4), %edi
        call    scriviint
        incl    i(%rbp)             # i++
        jmp     rip2
avan2:  call    nuovalinea

        movl    $0, %eax

        leave                       # epilogo
        ret
```

Esempio 6: parametro array (3)

```
# programma array, file es6b.s
# registri per i parametri: rdi, rsi
.text
.set  i, -24
.set  a, -16      # puntatore al  primo elemento
.set  n, -8
.global raddoppia
raddoppia:
      pushq %rbp              # prologo
      movq  %rsp, %rbp
      subq   $24, %rsp
      # i: 4 byte, a: 8 byte (allineato), n: 4 byte

      movq  %rdi, a(%rbp)     # a (puntatore)
      movl   %esi, n(%rbp)

      movl   $0, i(%rbp)      # i = 0
rip: movl   n(%rbp), %edi
      cmpl    %edi, i(%rbp)
      jge     fine            # salta se i >= n
```

```
      movq     a(%rbp), %rdi
      movslq   i(%rbp), %rsi
      movl     (%rdi, %rsi, 4), %edx
      addl     %edx, %edx
      movl     %edx, (%rdi, %rsi, 4)
      incl     i(%rbp)
      jmp      rip

fine: leave    # epilogo
      ret
```

Esempio 7: array locale e globale (1)

```
// programma arrayLocGlob, file es7a.cpp
#include"servi.cpp"
int alfa[5];
extern "C" void addar(int a[], int b[], int n);
int main()
{     int beta[5]; int i;
      for (i=0; i<5; i++) alfa[i]= leggiint();
      for (i=0; i<5; i++) beta[i]= leggiint();
      addar(alfa, beta, 5);
      for (i=0; i<5; i++) scriviint(alfa[i]);
      nuovalinea();
      return 0;
};

// programma arrayLocGlob, file es7b.cpp
extern "C" void addar(int a[], int b[], int n)
{     int i;
      for (i=0; i<n; i++) a[i]= a[i]+b[i];
};
```

Esempio 7: array locale e globale (2)

```
# programma arrayLocGlob, file es9a.s
# registri per i parametri: rdi, rsi, rdx
.include "servi.s"

.data
.global alfa
alfa:  .fill    5, 4

.text
#.extern addar
.global main
.set  beta, -24
.set  i,  -4
main:
      pushq %rbp            # prologo
      movq  %rsp, %rbp
      subq   $24, %rsp
```

```
          movl  $0, i(%rbp)      # i = 0
rip1:     cmpl  $5, i(%rbp)
          jge    avan1           # salta se i >= 5
          call   leggiint
          leaq  alfa(%rip), %rsi
          # array globale, indirizzo relativo a rip
          movslq   i(%rbp), %rdi
          movl %eax, (%rsi,%rdi, 4)
          # intero letto in alfa[i]
          incl   i(%rbp)         # i++
          jmp  rip1
avan1: movl  $0, i(%rbp)
rip2:     cmpl  $5, i(%rbp)
          jge    avan2           # salta se i >= 5
          call   leggiint
          movslq   i(%rbp), %rdi
          movl  %eax, beta(%rbp,%rdi,4)
          # array locale, beta non e' un indirizzo,
          # ma un displacement numerico (.set)
          incl    i(%rbp)        # i++
          jmp   rip2
```

Esempio 7: array locale e globale (3)

```
# programma arrayLocGlob, file es9a.s, continua …
avan2:   leaq   alfa(%rip), %rdi      # array globale, indirizzo relativo a rip
         leaq   beta(%rbp), %rsi      # array locale: numero (.set) relativo a rbp
         movl   $5, %edx
         call   addar

         movl  $0, i(%rbp)
 rip3:   cmpl  $5, i(%rbp)
         jge    avan3                 # salta se i >= 5
         leaq   alfa(%rip), %rax      # arry globale
         movslq  i(%rbp), %rsi
         movl   (%rax,%rsi, 4), %edi
         call   scriviint
         incl   i(%rbp)               # i++
         jmp    rip3
avan3:   call   nuovalinea

         movl   $0, %eax

         leave                        # epilogo
         ret
```

Esempio 7: array locale e globale (4)

```
# programma arrayLocGlob, file es7b.s
# registri per i parametri: rdi, rsi, rdx
.text
.set  i, -32
.set  a, -24
.set  b, -16
.set  n, -8
.global addar
addar:
      pushq %rbp              # prologo
      movq  %rsp, %rbp
      subq  $32, %rsp         # allineamento

      movq  %rdi, a(%rbp)
      movq  %rsi, b(%rbp)
      movl  %edx, n(%rbp)

      movl  $0, i(%rbp)       # i = 0
rip:  movl  n(%rbp), %edx
      cmpl  %edx, i(%rbp)
      jge   fine              # salta se i >= n

      movslq  i(%rbp),%rcx
      movq    a(%rbp), %rdi
      movq    b(%rbp), %rsi
      movl    (%rdi, %rcx, 4), %eax
      addl    (%rsi, %rcx, 4), %eax
      movl    %eax, (%rdi, %rcx, 4)
      incl    i(%rbp)
      jmp     rip

fine: leave           # epilogo
      ret
```

Esempio 8: scambio (1)

```
// programma scambio, file es8a.cpp
#include"servi.cpp"
extern "C" void sca(int& r1, int& r2);
int main()
{     int a, b;
      a = leggiint(); b = leggiint();
      sca(a, b);
      scriviint(a); scriviint(b); nuovalinea();
      return 0;
};

// programma scambio, file es8b.cpp
extern "C" void sca(int& r1, int& r2)
{     int lav;
      lav = r1; r1 = r2; r2 = lav;  // non scambia r1 con r2, ma gli interi riferiti da r1 ed r2
};
```

Esempio 8: scambio (2)

```
# programma scambio, file es8a.s
# registri per i parametri: rdi, rsi
.include  "servi.s"
.text
# .extern sca
.global main
.set  a, -8
.set  b, -4
main:
      pushq %rbp              # prologo
      movq  %rsp, %rbp
      subq   $8, %rsp
      # a: 4 byte, b: 4 byte

      call     leggiint
      movl   %eax, a(%rbp)
      call     leggiint
      movl   %eax, b(%rbp)
```

```
      leaq    a(%rbp), %rdi       # parametro &a
      leaq    b(%rbp), %rsi       # parametro &b
      call     sca

      movl    a(%rbp), %edi
      call     scriviint
      movl     b(%rbp), %edi
      call     scriviint
      call     nuovalinea

      movq    $0, %rax
      leave                        # epilogo
      ret
```

Esempio 8: scambio (3)

```
# programma scambio, file es8b.s
# registri per i parametri: rdi, rsi
.text
.global sca
.set  lav, -24
.set  r1, -16
.set  r2, -8
sca:
      pushq %rbp              # prologo
      movq  %rsp, %rbp
      subq   $24, %rsp
      # lav: 4 byte, r1: 8 byte, r2: 8 byte

      movq  %rdi, r1(%rbp)
      movq  %rsi, r2(%rbp)
      movq  r1(%rbp), %rax    # lav = *r1
      movl   (%rax), %eax
      movl   %eax, lav(%rbp)
```

```
      movq   r1(%rbp), %rax   # *r1 = *r2
      movq   r2(%rbp), %rdx
      movl    (%rdx), %edx
      movl     %edx, (%rax)

      movq    r2(%rbp), %rax  # *r2 = lav
      movl    lav(%rbp), %edx
      movl    %edx, (%rax)

      leave                           # epilogo
      ret
```

Esempio 9 : massimo riferito (1)

```
// programma massimo, file es9a.cpp
#include"servi.cpp"
extern "C" int& massi(int& ra, int& rb);
int main()
{     int a, b;
      a = leggiint(); b = leggiint();
      massi(a, b) = 0;
      scriviint(a); scriviint(b); nuovalinea();
};

// programma massimo, file es9b.cpp
extern "C" int& massi(int& ra, int& rb)
{     if (ra >= rb) return ra; return rb;
      // i riferimenti vengono automaticamente dereferenziati
      // la funzione restituisce il riferimento alla variabile riferita di valore massimo
};
```

Esempio 9 : massimo riferito (2)

```
# programma massimo, file es9a.s
# registri per i parametri: rdi, rsi,
# registri per il  risultato: rax
.include "servi.s"
.text
# .extern massi
.global main
.set  a, -8
.set  b, -4
main:
      pushq %rbp            # prologo
      movq  %rsp, %rbp
      subq   $8, %rsp
      # a: 4 byte, b: 4 byte

      call     leggiint
      movl   %eax, a(%rbp)
      call     leggiint
      movl   %eax, b(%rbp)
```

```
      leaq   a(%rbp), %rdi    # a &
      leaq   b(%rbp), %rsi    # b &
      call    massi
      movl  $0, (%rax)        # massi() = 0;

      movl   a(%rbp), %edi
      call    scriviint
      movl   b(%rbp), %edi
      call   scriviint
      call    nuovalinea

      movl   $0, %eax

      leave                   # epilogo
      ret
```

Esempio 9 : massimo riferito (3)

```
# programma massimo, file es9b.s
# registri per i parametri: rdi, rsi
# registri per il  risultato: rax,
.text
.global massi
.set  ra, -16
.set  rb, -8
massi:
      pushq %rbp                  # prologo
      movq  %rsp, %rbp
      subq   $16, %rsp
      # ra: 8 byte, rb: 8 byte

      movq  %rdi, ra(%rbp)
      movq  %rsi, rb(%rbp)

      movq  ra(%rbp), %rax   # ra
      movl   (%rax), %edi      # (ra)
      movq  rb(%rbp), %rax   # rb
      movl   (%rax), %esi      # (rb)
      cmpl   %edi, %esi
      jl        oltre               # salta se (rb)<(ra)
```

```
                movq   rb(%rbp),%rax
                jmp      avanti
oltre:          movq   ra(%rbp), %rax
                # rax: puntatore alla variabile
                # di valore maggiore

avanti:         leave          # epilogo
                ret

# risultato in rax: ra oppure rb, essendo la
# funzione di tipo riferimento int &
# se fosse stata di tipo int avremmo avuto:
# la successiva istruzione
#  movl      (%rax), %rax
```

Parametri e risultato di un tipo struttura (o unione)

- **Caso semplice (preso in esame nell'esempio successivo):**
  - **parametri e risultato valore di un tipo struttura, con numero di byte del tipo non superiore a 16:**
  - **per ogni parametro possono essere utilizzati fino a 2 registri consecutivi liberi, fra RDI, RSI, RDX, RCX, R8, R9;**
    - **il numero totale dei registri utilizzati per i parametri non deve comunque superare 6.**
  - **per il risultato può essere utilizzata, in tutto o in parte, la coppia di registri RAX-RDX.**
- **Parametri e risultato di un tipo riferimento di struttura:**
  - **il numero di byte richiesto dal tipo riferito può essere qualsivoglia;**
  - **un suo riferimento occupa sempre un solo registro.**
- **Allineamenti di strutture nel record di attivazione:**
  - **l'ambiente locale occupa un numero di byte multiplo di 8 byte;**
  - **le strutture, per semplicità, si allineano di solito a multipli di 8 byte, anche se l'allineamento minimo da rispettare è determinato dal campo avente vincoli maggiori.**

Esempio 10: parametro struttura (1)

```
// programma struttura, file es10a.cpp
#include"servi.cpp"
struct s { int n1; char c; int n2; };
extern "C" s leggis()
{     s ss;
      ss.n1 = leggiint(); ss.c = leggichar();
                ss.n2 = leggiint();
      return ss;            // in rax, edx
};
extern "C" void scrivis(s ss)
{     scriviint(ss.n1); scrivichar(ss.c);
                scriviint(ss.n2); nuovalinea();
};
extern "C" s fai(s st);
int main()
{     s st1, st2;
      st1 = leggis();
      st2 = fai(st1);
      scrivis(st2);
      return 0;
};
```

```
// programma struttura, file es10b.cpp
struct s { int n1; char c; int n2; };
extern "C" s fai(s st)
{     s ss;
      ss.n1 = st.n1+ 5; ss.c = st.c + 1;
      ss.n2 = st.n2 + 10;
      return ss;
}
```

Esempio 10: parametro struttura (2)

```
# programma struttura, file es10a.s
# registri per il  risultato di legis: rax, rdx
.include "servi.s"
.text
# ss.n1: ss(%rbp), ss.c: ss+4(%rbp)
# ss.n2: ss+8(%rbp)

.set  ss, -16
leggis:
      pushq %rbp          # prologo
      movq  %rsp, %rbp
      subq   $16, %rsp

      call     leggiint
      movl   %eax, ss(%rbp)
      call     leggichar
      movb  %al, ss+4(%rbp)
      call     leggiint
      movl   %eax, ss+8(%rbp)

      movq  ss(%rbp), %rax   # return ss
      movl   ss+8(%rbp), %edx
      leave
      ret
```

```
# registri per il parametro di scrivis: rdi, rsi
.set   ss, -16
 scrivis:
      pushq  %rbp
      movq    %rsp, %rbp
      subq    $16, %rsp

      movq    %rdi, ss(%rbp)       # ss.n1 e ss.c
      movl     %esi, ss+8(%rbp)   # ss.n2

      movl      ss(%rbp), %edi
      call       scriviint
       movb     ss+4(%rbp), %dil
       call       scrivichar
       movl      ss+8(%rbp), %edi
       call       scriviint
       call       nuovalinea

       leave                  # epilogo
       ret
```

Esempio 10: parametro struttura (3)

```
# programma struttura, file es10a.s, continua
      …
# registri per il parametro di fai: rdi, rsi
# registri per il  risultato di fai: rax, rdx
#.extern fai
.global main
.set st1, -32
.set st2, -16
main:
      pushq %rbp                          # prologo
      movq  %rsp, %rbp
      subq   $32, %rsp
      # st1, st2: 16 byte ciascuna

      call     leggis
      movq  %rax, st1(%rbp)
      movl   %edx, st1+8(%rbp)
```

```
movq    st1(%rbp), %rdi
movl    st1+8(%rbp), %esi
call     fai
movq    %rax, st2(%rbp)   # risultato in  st2
movl    %edx, st2+8(%rbp)

movq     st2(%rbp), %rdi
movl     st2+8(%rbp), %esi
call     scrivis

movl     $0, %eax

leave                              # epilogo
ret
```

Esempio 10: parametro struttura (4)

```
# programma struttura, file es10b.s
# registri per il parametro: rdi, rsi
# registri per il  risultato: rax, rdx
.text
.global fai
.set ss, -32
.set st, -16
fai:
      pushq %rbp                    # prologo
      movq  %rsp, %rbp
      subq   $32, %rsp
      # var. ss: 16 byte, par. st: 16 byte,

      movq  %rdi, st(%rbp)
      movl   %esi, st+8(%rbp)

      movl   st(%rbp), %eax
      addl    $5, %eax
      movl   %eax, ss(%rbp)
      movb  st+4(%rbp), %al
      addb   $1, %al
      movb  %al, ss+4(%rbp)

      movl    st+8(%rbp), %eax
      addl    $10, %eax
      movl    %eax, ss+8(%rbp)

      movq   ss(%rbp), %rax
      movl    ss+8(%rbp), %edx

      leave                         # epilogo
      ret
```

Esempio 11: parametro riferimento di struttura (1)

```
// programma strutturaRif, file es11a.cpp
#include"servi.cpp"
struct s { int n1; char c; int n2; };
extern "C" s leggis()
{     s ss;
      ss.n1 = leggiint(); ss.c = leggichar();
            ss.n2 = leggiint();
      return ss;
};
extern "C" void scrivis(s ss)
{     scriviint(ss.n1); scrivichar(ss.c);
         scriviint(ss.n2); nuovalinea();
};
extern "C" void fair(s& ss);
int main()
{     s st;
      st = leggis();
      fair(st);
      scrivis(st);
      return 0;
};
```

```
// programma strutturaRif, file es11b.cpp
struct s { int n1; char c; int n2; };
extern "C" void fair(s& ss)
{               ss.n1 = ss.n1+5; ss.c = ss.c+1;
                ss.n2 = ss.n2 + 10;
}
```

Esempio 11: parametro riferimento di struttura (2)

```
# programma strutturaRif, file es11a.s
.include "servi.s"
.text
# .extern fair
leggis: …
scrivis: …

# registro per il parametro: rdi
.global main
.set st, -12
main:
      pushq %rbp              # prologo
      movq  %rsp, %rbp
      subq  $16, %rsp       # allineamento

      call    leggis
      movq  %rax, st(%rbp)
      movl  %edx, st+8(%rbp)

      leaq    st(%rbp), %rdi   # st&
      call    fair
```

```
      movq      st(%rbp), %rdi
      movl      st+8(%rbp), %esi
      call      scrivis

      movl      $0, %eax

      leave                   # epilogo
      ret
```

Esempio 11: parametro riferimento di struttura (3)

```
# programma strutturaRif, file es11b.s
# registro per il parametro: rdi
.text
.global fair
.set ss, -8
fair:
      pushq %rbp                # prologo
      movq  %rsp, %rbp
      subq   $8, %rsp
      # parametro ss (riferimento): 8 byte

      movq  %rdi, ss(%rbp)    # rdi: riferimento st&
      movq  ss(%rbp), %rdi
      addl    $5, (%rdi)
      incb    4(%rdi)
      addl    $10, 8(%rdi)

      leave                     # epilogo
      ret
```

```
# programma strutturaRif, file es11b.s
# versione semplificata
# rdi: registro per il parametro
# rdi: contiene il riferimento ss
.text
.global fair
fair:
            addl        $5, (%rdi)
            incb        4(%rdi)
            addl        $10, 8(%rdi)

            ret
```

Risultato: struttura con più di 16 byte

- **Caso di un risultato di un tipo struttura che occupa più di 16 byte (i parametri valore, se di un tipo struttura, occupano invece un numero di byte inferiore a 16).**
- **Azioni del chiamante:**
  - **risultato (struttura che occupa più di 16 byte):**
    - **utilizza per la struttura risultato una variabile di lavoro (fra le variabili locali);**
    - **tramite il registro RDI, trasmette al chiamato un primo parametro aggiuntivo costituito dall'indirizzo della variabile di lavoro.**
- **Azioni del chiamato:**
  - **effettua le dovute elaborazioni, tenendo conto che i registri per i parametri cominciano da RSI;**
  - **lascia il risultato all'indirizzo specificato dal chiamante col primo parametro aggiuntivo (contenuto in RDI);**
  - **ricopia tale indirizzo in RAX (ai fini di una verifica da parte del chiamante).**

Esempio 12: risultato struttura lunga (1)

```
// programma strutturaLunga, file
      es12a.cpp
#include"servi.cpp"
struct s { int n1; int n2; char a[10]; };
extern "C" s fstruct(int a, char c);
extern "C" void scriviris(s& ss)
{     int i;
      scriviint(ss.n1); scriviint(ss.n2);
      for (i=0; i<10; i++) scrivichar(ss.a[i]);
      nuovalinea();
}
int main()
{     s sa;
      sa = fstruct(5, 'a');
      scriviris(sa);
      return 0;
      # la struttura non viene letta,
      # ma restituita da fstruct()
};
```

```
// programma strutturaLunga, file es12b.cpp
struct s { int n1; int n2; char a[10]; };
extern "C" s fstruct(int a, char c)
{       int i; s st;
        st.n1 = a; st.n2  = 2*a;
        for (i=0; i<10; i++) st.a[i] = c+i;
        return st;
}
```

Esempio 12: risultato struttura lunga (2)

```
# programma strutturaLunga, file es12a.s
.include "servi.s"
.text
# registro per il  parametro: rdi
.set  i, -16
.set  ss, -8     # parametro riferimento
scriviris:  pushq %rbp     # prologo
        movq   %rsp, %rbp
        subq    $16, %rsp

        movq   %rdi, ss(%rbp)
        movq    ss(%rbp), %rax
        movl    (%rax), %edi            # ss.n1
        call      scriviint
        movq   ss(%rbp), %rax
        movl    4(%rax), %edi           # ss.n2
        call      scriviint
        movl    $0, i(%rbp)             # i
ciclo: cmpl   $10, i(%rbp)
        jge       finec
        movq    ss(%rbp), %rax
        movslq i(%rbp), %rcx
        movb    8(%rax,%rcx), %dil  # ss.a[i]
      call     scrivichar  #dil parte di rdi
          incl  i(%rbp)
          jmp  ciclo
finec:  call   nuovalinea
          leave                         # epilogo
          ret

 .global main
.set  sa, -48
.set  lav, -24
main:
          pushq   %rbp              # prologo
          movq    %rsp, %rbp
          subq    $48, %rsp
          leaq    lav(%rbp), %rdi # par.aggiuntivo
          movl    $5, %esi          # parametro 5 II reg
          movb    $'a', %dl         # parametro 'a' III reg
          call      fstruct
          movq    lav(%rbp), %rax # lav in sa, 18 byte
          movq    %rax, sa(%rbp)
          movq    lav+8(%rbp), %rax
          movq    %rax, sa+8(%rbp)
          movw    lav+16(%rbp), %ax
          movw    %ax, sa+16(%rbp)
          leaq    sa(%rbp), %rdi
          call      scriviris

          movl    $0, %eax
          leave                             # epilogo
          ret
```

Esempio 12: risultato struttura lunga (3)

```
# programma strutturaLunga, file es12b.s
# registri per i parametri: rdi, rsi, rdx
# di e dil appartengono  a rdi; dx e dl a rdx
.text
.global fstruct
.set  i, -40
.set  st, -32          # 18 byte, allineato a 24
.set  a, -8
.set  c, -4
fstruct:
      pushq %rbp                  # prologo
      movq  %rsp, %rbp
      subq   $40, %rsp

      movl   %esi, a(%rbp)       # II registro
      movb  %dl, c(%rbp)         # III registro

      movl   a(%rbp), %eax       # st.n1 = a
      movl   %eax, st(%rbp)

      movl    a(%rbp), %eax      # st.n2 = 2*a
      addl    a(%rbp), %eax
      movl    %eax, st+4(%rbp)
```

```
            movl     $0, i(%rbp)        # st.a[i]=c+i
ciclo:      cmpl     $10, i(%rbp)
            jge      finec
            movzbl   c(%rbp), %eax
            addl     i(%rbp), %eax
            movslq   i(%rbp), %rcx
            movb     %al, st+8(%rbp,%rcx)
            incl      i(%rbp)
            jmp      ciclo

finec:      movq     %rdi, %rax    # rdi: ind. risultato
            movq     st(%rbp), %rdi    # ritorna st
            movq     %rdi, (%rax)
            movq     st+8(%rbp), %rdi
            movq     %rdi, 8(%rax)
            movw     st+16(%rbp), %di
            movw     %di, 16(%rax)

            leave  # epilogo (ind. risultato già in rax)
            ret
```

Esempio 12: risultato struttura lunga (4)

- **Regola generale:**
  - **non è detto che il risultato di una funzione debba essere assegnato a una variabile;**
  - **esempi:**

```
scriviris(fstruct(5,'a');
sa = fstruct(5,'a') + fstruct(10,'m'); // operatore + per le strutture: ridefinito
```

- **Ottimizzazione possibile in caso di assegnamento:**
  - **trasmettere come indirizzo del risultato, invece che quello di una variabile ausiliaria *lav*, quello della variabile *sa* a cui viene assegnato il risultato.**

Esempio 13: parametro riferimento di struttura lunga (1)

```
// programma strutturaLungaRif, file
      es13a.cpp
#include"servi.cpp"
struct s { int n1; int n2; char a[10]; };
extern "C" void fstructr(s& st, int a, char c);
extern "C" void scriviris(s& ss)
{     int i;
      scriviint(ss.n1); scriviint(ss.n2);
      for (i=0; i<10; i++) scrivichar(ss.a[i]);
      nuovalinea();
}
int main()
{     s sa;
      fstructr(sa, 5, 'a');
      scriviris(sa);
      return 0;
};
```

```
// programma strutturaLungaRif, file es13b.cpp
struct s { int n1; int n2; char a[10]; };
extern "C" void fstructr(s& st, int a, char c)
{     int i;
      st.n1 = a; st.n2  = 2*a;
      for (i=0; i<10; i++) st.a[i] = c+i;
}
```

- **Con questa soluzione:**
  - **non esistono parametri aggiuntivi;**
  - **il primo parametro effettivo (riferimento) è visibile anche in C++;**
  - **in Assembler, si deve semplicemente tradurre quanto scritto in C++.**

Esempio 13: parametro riferimento di struttura lunga (2)

```
# programma strutturaLungaRif, file
      es13a.s
.include "servi.s"
.text
# registro per il  parametro: rdi
.set   i, -16
.set   ss, -8        # parametro riferimento
      (puntatore)
scriviris: pushq %rbp              # prologo
        movq        %rsp, %rbp
        subq        $16, %rsp
        movq        %rdi, ss(%rbp)
        movq        ss(%rbp), %rax
        movl        (%rax), %edi         # (*ss).n1
        call        scriviint
        movq        ss(%rbp), %rax
        movl        4(%rax), %edi        # (*ss).n2
        call        scriviint
        movl        $0, i(%rbp)                 # i
ciclo: cmpl         $10, i(%rbp)
        jge         finec
                    movq      ss(%rbp), %rax
                    movslq    i(%rbp), %rcx
                    movb      8(%rax,%rcx), %dil #
(*ss).a[i]
                    call                    scrivichar
                    incl      i(%rbp)
                    jmp       ciclo
finec:  call                  nuovalinea

                    leave
                  # epilogo
                    ret
```

Esempio 13: parametro riferimento di struttura lunga (3)

```
# programma strutturaLungaRif, file es13a.s, continua
.global main
.set   sa, -24
main:
        pushq       %rbp        # prologo
        movq        %rsp, %rbp
        subq        $24, %rsp          # sa: 18 byte allineati

        leaq        sa(%rbp), %rdi  # I par &lav:
        movl        $5, %esi           # II parametro 5
        movb        $'a', %dl          # III parametro 'a'
        call        fstructr
        leaq        sa(%rbp), %rdi
        call        scriviris
        movl        $0, %eax

        leave                   # epilogo
        ret
```

Esempio 13: parametro riferimento di struttura lunga (4)

```
# programma strutturaLungaRif, file es13b.s
# registri per i parametri: rdi, rsi, rdx
.text
.global fstructr
.set   i, -40
.set   st, -32
.set   a, -8
.set   c, -4
fstructr:
        pushq     %rbp              # prologo
        movq      %rsp, %rbp
        subq      $40, %rsp         # i, st, a, c

        movq      %rdi, st(%rbp)    # I par. (s*)
        movl      %esi, a(%rbp)     # II par. (int)
        movb      %dl, c(%rbp)      # III par. (char)

        movl      a(%rbp), %eax     # (*st).n1 = a
        movq      st(%rbp), %rbx
        movl      %eax, (%rbx)
                  movl      a(%rbp), %eax  #
(*st).n2 = 2*a
                  addl      a(%rbp), %eax
                  movq  st(%rbp), %rbx
                  movl      %eax, 4(%rbx)

                  movl      $0, i(%rbp)
# (*st).a[i]=c+i
ciclo: cmpl $10, i(%rbp)
                  jge       finec
                  movzbl c(%rbp), %eax
                  addl      i(%rbp), %eax
                  movslq i(%rbp), %rcx
                  movq  st(%rbp), %rbx
                  movb  %al, 8(%rbx,%rcx)
                  incl      i(%rbp)
                  jmp       ciclo

finec: leave
# epilogo
                   ret
```